Matrices

Exercice 1

```
> with(LinearAlgebra):
```

1.

```
> A:=<<1,2,2>|<1,1,2>|<1,5,6>|<1,3,4>>;
  B:=<<1,1,1,1>|<1,2,0,2>|<1,1,3,0>>;
```

$$A := \begin{bmatrix} 1 & 1 & 1 & 1 \\ 2 & 1 & 5 & 3 \\ 2 & 2 & 6 & 4 \end{bmatrix}$$

$$B := \begin{bmatrix} 1 & 1 & 1 \\ 1 & 2 & 1 \\ 1 & 0 & 3 \\ 1 & 2 & 0 \end{bmatrix}$$

2.

```
> A[2,3];
```

$$5$$

3.

```
> C:=A.B;
  B.A;
```

$$C := \begin{bmatrix} 4 & 5 & 5 \\ 11 & 10 & 18 \\ 14 & 14 & 22 \end{bmatrix}$$

$$\begin{bmatrix} 5 & 4 & 12 & 8 \\ 7 & 5 & 17 & 11 \\ 7 & 7 & 19 & 13 \\ 5 & 3 & 11 & 7 \end{bmatrix}$$

4.

```
> A.<1,2,3,4>;
  A.<1,0,0,0>;
  A.<0,1,0,0>;
  A.<0,0,1,0>;
  A.<0,0,0,1>;
```

$$\begin{bmatrix} 10 \\ 31 \\ 40 \end{bmatrix}$$

$$\begin{bmatrix} 1 \\ 2 \\ 2 \end{bmatrix}$$

$$\begin{bmatrix} 1 \\ 1 \\ 2 \end{bmatrix}$$

$$\begin{bmatrix} 1 \\ 5 \\ 6 \end{bmatrix}$$

$$\begin{bmatrix} 1 \\ 3 \\ 4 \end{bmatrix}$$

5.

```
> C1:=Transpose(A.B);
  C2:=Transpose(B).Transpose(A);
```

$$C1 := \begin{bmatrix} 4 & 11 & 14 \\ 5 & 10 & 14 \\ 5 & 18 & 22 \end{bmatrix}$$

$$C2 := \begin{bmatrix} 4 & 11 & 14 \\ 5 & 10 & 14 \\ 5 & 18 & 22 \end{bmatrix}$$

Remarquons qu'il n'est pas possible de tester l'égalité de deux matrices de la façon suivante :

```
> evalb(C1=C2);
```

$$\mathit{false}$$

6.

```
> Determinant(C);
```

$$-8$$

Le determinant est non nul. La matrice C est inversible.

```
> MatrixInverse(C);
```

$$\begin{bmatrix} 4 & 5 & -5 \\ \frac{-5}{4} & \frac{-9}{4} & \frac{17}{8} \\ \frac{-7}{4} & \frac{-7}{4} & \frac{15}{8} \end{bmatrix}$$

7.

```
> ColumnSpace(A);
```

$$\left[\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix}\right]$$

On obtient la base canonique. X->AX est donc une application surjective.

8.

```
> NullSpace(A);
```

$$\left\{\begin{bmatrix} 0 \\ \frac{-1}{2} \\ \frac{-1}{2} \\ 1 \end{bmatrix}\right\}$$

9.(a)

```
> X:=LinearSolve(A,<1,2,3>,free=x);
```

$$X := \begin{bmatrix} 0 \\ \frac{1}{2} + x_3 \\ x_3 \\ \frac{1}{2} - 2\,x_3 \end{bmatrix}$$

9.(b)

```
> x0:=solve(X[4]=1);
  subs(x[3]=x0,X);
```

$$x0 := \frac{-1}{4}$$

$$\begin{bmatrix} 0 \\ \frac{1}{4} \\ \frac{-1}{4} \\ 1 \end{bmatrix}$$

Exercice 1

```
> restart;
  with(LinearAlgebra):
  A:=<<1,-3,0,-1>|<1,1,1,0>|<0,4,1,1>|<2,0,3,1>>;
  I4:=IdentityMatrix(4);
```

$$A := \begin{bmatrix} 1 & 1 & 0 & 2 \\ -3 & 1 & 4 & 0 \\ 0 & 1 & 1 & 3 \\ -1 & 0 & 1 & 1 \end{bmatrix}$$

$$I4 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

1.(a)

```
> P:=X->Determinant(X*I4-A);
  P(X);
```

$$P := X \rightarrow LinearAlgebra\text{:-}Determinant(X\ I4 - A)$$

$$X^4 - 4\,X^3 + 4\,X^2$$

1.(b)

```
> solve(P(X),X);
```

$$0, 0, 2, 2$$

Le polynôme P est appelélepolynôme caractéristique de A.
Si r_0 est une racine de P, alors A-r_0I_4 a un noyau non réduit à {0}. On peut donc trouver X non nul tel que (A-r_0I_4).X=AX-r_0X et donc A.X=r_0X

2.(a)

```
> B0:=NullSpace(A);
  u0:=B0[1];
```

$$B0 := \left\{ \begin{bmatrix} 1 \\ -1 \\ 1 \\ 0 \end{bmatrix} \right\}$$

$$u0 := \begin{bmatrix} 1 \\ -1 \\ 1 \\ 0 \end{bmatrix}$$

2.(b)

```
> v0:=LinearSolve(A,u0,free=x);
```

$$v0 := \begin{bmatrix} x_3 + \frac{1}{3} \\ -x_3 \\ x_3 \\ \frac{1}{3} \end{bmatrix}$$

LinearSolve nous la forme générale de la solution du problème. Nous faisons un choix arbitraire pour v_0

```
> v0:=subs(x[3]=0,v0);
```

$$v0 := \begin{bmatrix} \frac{1}{3} \\ 0 \\ 0 \\ \frac{1}{3} \end{bmatrix}$$

3.(a)

```
> B2:=NullSpace(A-2*I4);
  u2:=B2[1];
```

$$B2 := \left\{ \begin{bmatrix} 1 \\ 1 \\ 1 \\ 0 \end{bmatrix} \right\}$$

$$u2 := \begin{bmatrix} 1 \\ 1 \\ 1 \\ 0 \end{bmatrix}$$

3.(b)

On a : A.v_2=u_2+2.v_2 <=> A.v_2-2.v_2=u_2 <=> (A-2.I_4).v_2=u_2

```
> v2:=LinearSolve(A-2*I4,u2,free=x);
```

$$v2 := \begin{bmatrix} x_3 - \frac{1}{3} \\ x_3 \\ x_3 \\ \frac{1}{3} \end{bmatrix}$$

A nouveau, nous avons la forme générale de la solution. Nous faisons un choix arbitraire pour v_2.

```
> v2:=subs(x[3]=0,v2);
```

$$v2 := \begin{bmatrix} \frac{-1}{3} \\ 0 \\ 0 \\ \frac{1}{3} \end{bmatrix}$$

4.(a)

```
> P:=<u0|v0|u2|v2>;
  Determinant(P);
```

$$P := \begin{bmatrix} 1 & \frac{1}{3} & 1 & \frac{-1}{3} \\ -1 & 0 & 1 & 0 \\ 1 & 0 & 1 & 0 \\ 0 & \frac{1}{3} & 0 & \frac{1}{3} \end{bmatrix}$$

$$\frac{4}{9}$$

Le déterminant de P est non nul. Donc, C est une base.

4.(b)

On applique la formule de changement de bases.

```
> B:=MatrixInverse(P).A.P;
```

$$B := \begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 2 & 1 \\ 0 & 0 & 0 & 2 \end{bmatrix}$$

4.(c)

```
> seq(B^n,n=1..10);
```

$$\begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 2 & 1 \\ 0 & 0 & 0 & 2 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 4 & 4 \\ 0 & 0 & 0 & 4 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 8 & 12 \\ 0 & 0 & 0 & 8 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 16 & 32 \\ 0 & 0 & 0 & 16 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 32 & 80 \\ 0 & 0 & 0 & 32 \end{bmatrix},$$

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 64 & 192 \\ 0 & 0 & 0 & 64 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 128 & 448 \\ 0 & 0 & 0 & 128 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 256 & 1024 \\ 0 & 0 & 0 & 256 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 512 & 2304 \\ 0 & 0 & 0 & 512 \end{bmatrix},$$

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 1024 & 5120 \\ 0 & 0 & 0 & 1024 \end{bmatrix}$$

On peut constater de les coefficients (3,3) et (4,4) sont égaux à 2^n et que le coefficient (3,4) vaut n.2^(n-1).

Donc, si n>1, alors les autres coefficients sont tous nuls et donc B^n a la forme suivante :

```
> Bn:=<<0,0,0,0>|<0,0,0,0>|<0,0,2^n,0>|<0,0,n*2^(n-1),2^n>>;
```

$$Bn := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 2^n & n\,2^{(n-1)} \\ 0 & 0 & 0 & 2^n \end{bmatrix}$$

Les cas n=0 et n=1 sont triviaux. Nous les traiterons plus dorénavant.

4.(d)

On applique la formule de changement de base :

B^n=P^(-1).A^n.P

Donc : A^n=P.B^n.P^(-1)

```
> An:=P.Bn.MatrixInverse(P);
```

$$An := \begin{bmatrix} -\frac{3\,n\,2^{(n-1)}}{2} + \frac{2^n}{2} & \frac{2^n}{2} & \frac{3\,n\,2^{(n-1)}}{2} & \frac{3\,n\,2^{(n-1)}}{2} - \frac{2^n}{2} \\ -\frac{3\,n\,2^{(n-1)}}{2} & \frac{2^n}{2} & \frac{2^n}{2} + \frac{3\,n\,2^{(n-1)}}{2} & \frac{3\,n\,2^{(n-1)}}{2} \\ -\frac{3\,n\,2^{(n-1)}}{2} & \frac{2^n}{2} & \frac{2^n}{2} + \frac{3\,n\,2^{(n-1)}}{2} & \frac{3\,n\,2^{(n-1)}}{2} \\ -\frac{2^n}{2} & 0 & \frac{2^n}{2} & \frac{2^n}{2} \end{bmatrix}$$

Exercice 3

Tentons une approche directe :

```
> solve(A^2+A=<<1,1>|<1,1>>,A);
Error, (in solve) invalid arguments
```

Nous introduisons les coefficients de A :

```
> A:=<<a1,a2>|<b1,b2>>;
  B:=<<1,1>|<1,1>>;
  solve(A^2+A=<<1,1>|<1,1>>,{a1,a2,b1,b2});
```

$$A := \begin{bmatrix} a1 & b1 \\ a2 & b2 \end{bmatrix}$$

$$B := \begin{bmatrix} 1 & 1 \\ 1 & 1 \end{bmatrix}$$

```
Error, (in solve) invalid arguments
```

Nous stockons A^2+A dans C et nous en déduisons quatres équations. Nous résolvons le système associé.

```
> C:=A^2+A;
  Eq11:=C[1,1]=B[1,1];
  Eq12:=C[1,2]=B[1,2];
  Eq21:=C[2,1]=B[2,1];
  Eq22:=C[2,2]=B[2,2];
  sol:=solve({Eq11,Eq12,Eq21,Eq22},{a1,a2,b1,b2});
```

$$C := \begin{bmatrix} a1^2 + b1\ a2 + a1 & a1\ b1 + b1\ b2 + b1 \\ a2\ a1 + b2\ a2 + a2 & b1\ a2 + b2^2 + b2 \end{bmatrix}$$

$$Eq11 := a1^2 + b1\ a2 + a1 = 1$$

$$Eq12 := a1\ b1 + b1\ b2 + b1 = 1$$

$$Eq21 := a2\ a1 + b2\ a2 + a2 = 1$$

$$Eq22 := b1\ a2 + b2^2 + b2 = 1$$

$$sol := \{ b2 = 0, a2 = 1, a1 = 0, b1 = 1 \}, \{ b2 = \frac{-3}{2}, a2 = \frac{-1}{2}, a1 = \frac{-3}{2}, b1 = \frac{-1}{2} \},$$
$$\{ b2 = \frac{1}{2}, a2 = \frac{1}{2}, a1 = \frac{1}{2}, b1 = \frac{1}{2} \}, \{ a2 = -1, b2 = -1, a1 = -1, b1 = -1 \}$$

```
> for k from 1 to nops([sol]) do
      subs(sol[k],A);
  od;
```

$$\begin{bmatrix} 0 & 1 \\ 1 & 0 \end{bmatrix}$$

$$\begin{bmatrix} \frac{-3}{2} & \frac{-1}{2} \\ \frac{-1}{2} & \frac{-3}{2} \end{bmatrix}$$

$$\begin{bmatrix} \frac{1}{2} & \frac{1}{2} \\ \frac{1}{2} & \frac{1}{2} \end{bmatrix}$$

$$\begin{bmatrix} -1 & -1 \\ -1 & -1 \end{bmatrix}$$

Exercice 4

```
> A:=BandMatrix([1,0,1,0,a,0,1,0,1],4,6,outputoptions=[storage=rec
  tangular]);
```

$$A := \begin{bmatrix} a & 0 & 1 & 0 & 1 & 0 \\ 0 & a & 0 & 1 & 0 & 1 \\ 1 & 0 & a & 0 & 1 & 0 \\ 0 & 1 & 0 & a & 0 & 1 \\ 1 & 0 & 1 & 0 & a & 0 \\ 0 & 1 & 0 & 1 & 0 & a \end{bmatrix}$$

```
> seq(map(expand,A^n),n=1..3);
```

$$\begin{bmatrix} a & 0 & 1 & 0 & 1 & 0 \\ 0 & a & 0 & 1 & 0 & 1 \\ 1 & 0 & a & 0 & 1 & 0 \\ 0 & 1 & 0 & a & 0 & 1 \\ 1 & 0 & 1 & 0 & a & 0 \\ 0 & 1 & 0 & 1 & 0 & a \end{bmatrix}, \begin{bmatrix} a^2+2 & 0 & 2a+1 & 0 & 2a+1 & 0 \\ 0 & a^2+2 & 0 & 2a+1 & 0 & 2a+1 \\ 2a+1 & 0 & a^2+2 & 0 & 2a+1 & 0 \\ 0 & 2a+1 & 0 & a^2+2 & 0 & 2a+1 \\ 2a+1 & 0 & 2a+1 & 0 & a^2+2 & 0 \\ 0 & 2a+1 & 0 & 2a+1 & 0 & a^2+2 \end{bmatrix},$$

$$\begin{bmatrix} a^3+6a+2, 0, 3a^2+3+3a, 0, 3a^2+3+3a, 0 \\ 0, a^3+6a+2, 0, 3a^2+3+3a, 0, 3a^2+3+3a \\ 3a^2+3+3a, 0, a^3+6a+2, 0, 3a^2+3+3a, 0 \\ 0, 3a^2+3+3a, 0, a^3+6a+2, 0, 3a^2+3+3a \\ 3a^2+3+3a, 0, 3a^2+3+3a, 0, a^3+6a+2, 0 \\ 0, 3a^2+3+3a, 0, 3a^2+3+3a, 0, a^3+6a+2 \end{bmatrix}$$

On remarque que A^n est de la forme :

```
> An:=BandMatrix([b(n),0,b(n),0,c(n),0,b(n),0,b(n)],4,6,outputopti
  ons=[storage=rectangular]);
```

$$An := \begin{bmatrix} c(n) & 0 & b(n) & 0 & b(n) & 0 \\ 0 & c(n) & 0 & b(n) & 0 & b(n) \\ b(n) & 0 & c(n) & 0 & b(n) & 0 \\ 0 & b(n) & 0 & c(n) & 0 & b(n) \\ b(n) & 0 & b(n) & 0 & c(n) & 0 \\ 0 & b(n) & 0 & b(n) & 0 & c(n) \end{bmatrix}$$

```
> Bn:=An.A;
```

$$Bn := \begin{bmatrix} c(n)\,a+2\,b(n), 0, c(n)+b(n)\,a+b(n), 0, c(n)+b(n)\,a+b(n), 0 \\ 0, c(n)\,a+2\,b(n), 0, c(n)+b(n)\,a+b(n), 0, c(n)+b(n)\,a+b(n) \\ c(n)+b(n)\,a+b(n), 0, c(n)\,a+2\,b(n), 0, c(n)+b(n)\,a+b(n), 0 \\ 0, c(n)+b(n)\,a+b(n), 0, c(n)\,a+2\,b(n), 0, c(n)+b(n)\,a+b(n) \\ c(n)+b(n)\,a+b(n), 0, c(n)+b(n)\,a+b(n), 0, c(n)\,a+2\,b(n), 0 \\ 0, c(n)+b(n)\,a+b(n), 0, c(n)+b(n)\,a+b(n), 0, c(n)\,a+2\,b(n) \end{bmatrix}$$

An.A est bien de la même forme. Par le principe de récurrence, A^n est bien de cette forme.
On cherche des formules de récurrence pour les suites b(n) et c(n);

```
> Eq1:=b(n+1)=Bn[3,1];
  Eq2:=c(n+1)=Bn[1,1];
  sol:=rsolve({Eq1,Eq2,b(0)=0,c(0)=1},{b(n),c(n)});
```

$$Eq1 := b(n+1) = c(n) + b(n)\,a + b(n)$$

$$Eq2 := c(n+1) = c(n)\,a + 2\,b(n)$$

$$sol := \left\{\mathrm{b}(n) = -\frac{(a-1)^n}{3} + \frac{(a+2)^n}{3},\ \mathrm{c}(n) = \frac{2\,(a-1)^n}{3} + \frac{(a+2)^n}{3}\right\}$$

Ainsi, on obtient :

```
> A^n=subs(sol,An);
```

$$\begin{bmatrix} a & 0 & 1 & 0 & 1 & 0 \\ 0 & a & 0 & 1 & 0 & 1 \\ 1 & 0 & a & 0 & 1 & 0 \\ 0 & 1 & 0 & a & 0 & 1 \\ 1 & 0 & 1 & 0 & a & 0 \\ 0 & 1 & 0 & 1 & 0 & a \end{bmatrix}^n =$$

$$\left[\begin{array}{l}
\frac{2\,(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0,\, -\frac{(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0,\, -\frac{(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0 \\
0,\, \frac{2\,(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0,\, -\frac{(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0,\, -\frac{(a-1)^n}{3} + \frac{(a+2)^n}{3} \\
-\frac{(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0,\, \frac{2\,(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0,\, -\frac{(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0 \\
0,\, -\frac{(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0,\, \frac{2\,(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0,\, -\frac{(a-1)^n}{3} + \frac{(a+2)^n}{3} \\
-\frac{(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0,\, -\frac{(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0,\, \frac{2\,(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0 \\
0,\, -\frac{(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0,\, -\frac{(a-1)^n}{3} + \frac{(a+2)^n}{3},\, 0,\, \frac{2\,(a-1)^n}{3} + \frac{(a+2)^n}{3}
\end{array}\right]$$

```
>
```